

```

*****
** SAMPLE CONDITION HANDLING PROGRAM. **
** SHARE SESSION: EXPLOIT CONDITION HANDLING IN **
** LANGUAGE ENVIRONMENT **
** THOMAS PETROLINO **
** IBM LANGUAGE ENVIRONMENT **
** TAPETRO@US.IBM.COM **
** **
** THIS EXAMPLE IS MADE UP OF THREE PROGRAMS: **
** - ECHMAIN - THE MAIN ROUTINE **
** - ECHOUTBD - A SECOND ROUTINE CALLED BY ECHMAIN **
** - ECH911 - A USER-WRITTEN LE CONDITION HANDLER **
** **
** THIS EXAMPLE MUST BE COMPILED WITH OPTION SSRANGE IN **
** ORDER FOR IT TO DEMONSTRATE ALL OF ITS CAPABILITIES. **
*****
* -----

```

```

IDENTIFICATION DIVISION.
PROGRAM-ID. ECHMAIN.

```

```

DATA DIVISION.
WORKING-STORAGE SECTION.

```

```

*****
* AREAS FOR LE CONDITION HANDLER
*****

```

```

01 ERROR-INDICATOR PIC X EXTERNAL.
01 RECOVER-ADDR POINTER EXTERNAL.
01 MY-ABEND-TOKEN PIC S9(8) COMP.
01 PGMPTR PROCEDURE-POINTER.
01 SS-POINTER POINTER.
01 FEEDBACK.
   10 FB-SEV PIC S9(4) COMP.
   10 FB-MSG PIC S9(4) COMP.
   10 FB-CTL PIC X.
   10 FB-FAC PIC X(3).
   10 FB-ISINFO PIC S9(9) COMP.

```

```

*****
* END OF DATA VARIABLES FOR CONDITION HANDLER
*****

```

```

01 DIVISOR PIC 99.
01 Z PIC 99.
01 FP PROCEDURE-POINTER.

```

```

LINKAGE SECTION.
01 SIMPLE-STRUCTURE.
   10 SS-CHAR PIC X.

```

```

PROCEDURE DIVISION.
MAIN-PROG.

```

```

    DISPLAY 'ECHMAIN - STARTING'

```

```

*****
* REGISTER PROGRAM ECH911 AS A CONDITION HANDLER.
*****

```

```

    SET PGMPTR TO ENTRY 'ECH911'
    MOVE 0000 TO MY-ABEND-TOKEN
    CALL 'CEEHDLR' USING PGMPTR, MY-ABEND-TOKEN, FEEDBACK

```

```

IF FB-SEV = ZEROS
  DISPLAY 'ECHMAIN - ECH911 REGISTERED'
ELSE
  DISPLAY 'ECHMAIN - ECH911 REGISTRATION FAILED'
  DISPLAY 'FB-MSG = ' FB-FAC, FB-MSG
END-IF.

*****
* FORCE A DECIMAL DIVIDE EXCEPTION BY DIVIDING BY ZERO.
* WHEN ECH911 DETECTS THIS EXCEPTION, IT WILL REQUEST A
* RESUME TO THE NEXT INSTRUCTION FOLLOWING THE DIVIDE.
*****
MOVE 'N' TO ERROR-INDICATOR
DISPLAY "ECHMAIN - ATTEMPTING DIVIDE"
COMPUTE Z = 1 / DIVISOR.
IF ERROR-INDICATOR = 'Y'
  DISPLAY "ECHMAIN - RESUMED AFTER DECIMAL DIVIDE EXCP"
ELSE
  DISPLAY "ECHMAIN - DID NOT RESUME AFTER DECIMAL DIVIDE"
END-IF.

*****
* CALL PROGRAM ECHOUTBD, WHICH WILL FORCE A TABLE
* REFERENCE OUT OF BOUNDS ERROR. WHEN ECH911 DETECTS
* THIS EXCEPTION, IT WILL USE CEEMRCR TO MOVE THE RESUME
* CURSOR TO THE RETURN POINT FOLLOWING THE ECHOUTBD CALL.
*****
MOVE 'N' TO ERROR-INDICATOR
DISPLAY "ECHMAIN - CALLING ECHOUTBD"
CALL "ECHOUTBD".
IF ERROR-INDICATOR = 'Y'
  DISPLAY "ECHMAIN - RESUMED AFTER OOB ERROR"
ELSE
  DISPLAY "ECHMAIN - DID NOT RESUME AFTER OOB ERROR"
END-IF.

*****
* USE CEE3SRP TO SET UP A RESUME POINT AT THE NEXT
* STATEMENT AFTER THE CALL. FORCE A PROTECTION
* EXCEPTION. WHEN ECH911 DETECTS THIS CONDITION, IT WILL
* USE CEEMRCE TO MOVE THE RESUME CURSOR TO THE SAVED
* RESUME POINT.
*****
MOVE 'N' TO ERROR-INDICATOR
CALL 'CEE3SRP' USING RECOVER-ADDR, FEEDBACK.

IF ERROR-INDICATOR = 'N'
  DISPLAY "ECHMAIN - ATTEMPTING PROTECTION EXCP"
  SET SS-POINTER TO NULL
  SET ADDRESS OF SIMPLE-STRUCTURE TO SS-POINTER
  MOVE 'A' TO SS-CHAR
ELSE
  DISPLAY "ECHMAIN - RESUMED AFTER PROTECTION EXCP"
END-IF.

*****
* FORCE A WILD BRANCH TO ADDRESS ZERO, WHICH WILL RESULT

```

```

* IN AN OPERATION EXCEPTION. WHEN ECH911 DETECTS THIS
* EXCEPTION, IT WILL REQUEST PERCOLATION.
*****
    MOVE 'N' TO ERROR-INDICATOR
    DISPLAY "ECHMAIN  - ATTEMPTING WILD BRANCH"
    SET FP TO NULL.
    CALL FP.
* EXECUTION SHOULD NOT REACH HERE
    DISPLAY "ECHMAIN  - AFTER WILD BRANCH"

*****
* UNREGISTER CONDITION HANDLER ECH911
*****
    SET PGMPTR TO ENTRY 'ECH911'
    CALL 'CEEHDLU' USING PGMPTR, FEEDBACK
    IF FB-SEV = ZEROS
        DISPLAY 'ECHMAIN  - ECH911 UNREGISTERED'
    ELSE
        DISPLAY 'ECHMAIN  - ECH911 UNREGISTRATION FAILED'
        DISPLAY 'FB-MSG = ' FB-FAC, FB-MSG
    END-IF.

    DISPLAY 'ECHMAIN  - ENDING'
    GOBACK.
END PROGRAM "ECHMAIN".

* -----
IDENTIFICATION DIVISION.
PROGRAM-ID. ECHOUTBD.

DATA DIVISION.
WORKING-STORAGE SECTION.
01 ERROR-INDICATOR PIC X EXTERNAL.
77 J          PIC 9(4) USAGE COMP.
01 TABLE-X.
    02 SLOT PIC 9(4) USAGE COMP OCCURS 8 TIMES.
LINKAGE SECTION.
PROCEDURE DIVISION.
MAIN-PROG.
    DISPLAY 'ECHOUTBD - STARTING'

*****
* FORCE A TABLE REFERENCE OUT OF BOUNDS ERROR.
* WHEN ECH911 DETECTS THIS EXCEPTION, IT WILL USE
* CEEMRCR TO MOVE THE RESUME CURSOR TO THE RETURN POINT
* FOLLOWING THE ORIGINAL CALL TO THIS ROUTINE.
*****
    DISPLAY 'ECHOUTBD - ATTEMPTING AN OUT OF BOUNDS REFERENCE'
    MOVE 9 TO J.
    MOVE 1 TO SLOT (J).

* EXECUTION SHOULD NOT REACH HERE WHEN COMPILED WITH SSRANGE.
    IF ERROR-INDICATOR = 'Y'
        DISPLAY "ECHOUTBD - NOT COMPILED WITH SSRANGE"
    ELSE
        DISPLAY "ECHOUTBD - NOT COMPILED WITH SSRANGE"
    END-IF.

```

DISPLAY 'ECHOUTBD - ENDING'  
GOBACK.  
END PROGRAM "ECHOUTBD".

\* -----

IDENTIFICATION DIVISION.  
PROGRAM-ID. ECH911.  
ENVIRONMENT DIVISION.  
CONFIGURATION SECTION.  
INPUT-OUTPUT SECTION.

DATA DIVISION.  
WORKING-STORAGE SECTION.  
01 ERROR-INDICATOR PIC X EXTERNAL.  
01 RECOVER-ADDR POINTER EXTERNAL.  
01 FEEDBACK.  
    10 FB-SEV PIC S9(4) COMP.  
    10 FB-MSG PIC S9(4) COMP.  
    10 FB-CASE-SEV PIC X.  
    10 FB-FAC PIC X(3).  
    10 FB-ISINFO PIC S9(8) COMP.  
77 MOVE-TYPE PIC S9(9) COMP VALUE 0.

01 ROUTINE-NAME PIC X(80).  
01 MSG-DEST PIC S9(9) COMP VALUE +2.  
01 MSG-STR.  
    10 MSG-LENGTH PIC S9(4) COMP VALUE +120.  
    10 MSG-STRING PIC X(120).

LINKAGE SECTION.  
01 CURRENT-CONDITION.  
    10 FBCODE PIC X(8).  
        88 CEE341 VALUE X'00030C8159C3C5C5'.  
        88 CEE344 VALUE X'00030C8459C3C5C5'.  
        88 CEE34B VALUE X'00030C8B59C3C5C5'.  
        88 IGZ006 VALUE X'0003000659C9C7E9'.  
    10 PIC X(4).  
01 DATA-INFO PIC S9(8) COMP.  
01 RESULT-CODE PIC S9(9) COMP.  
    88 RESUME VALUE +10.  
    88 PERCOLATE VALUE +20.  
01 NEW-CONDITION PIC X(12).

PROCEDURE DIVISION USING CURRENT-CONDITION, DATA-INFO,  
RESULT-CODE, NEW-CONDITION.

100-PROGRAM-BEGIN.  
    DISPLAY 'ECH911 - ENTERED'

\*\*\*\*\*

\* USE CEE3GRN TO IDENTIFY THE PROGRAM THAT HAD THE  
\* CONDITION, AND WRITE IT OUT TO THE LE MSGFILE.  
\* THEN USE CEEMSG TO WRITE OUT THE CURRENT CONDITION

\*\*\*\*\*

CALL 'CEE3GRN' USING ROUTINE-NAME, FEEDBACK  
MOVE SPACES TO MSG-STRING  
STRING 'ECH911 - ROUTINE ' DELIMITED BY SIZE

ROUTINE-NAME DELIMITED BY ' '  
' ENCOUNTERED THE FOLLOWING CONDITION:'  
DELIMITED BY SIZE

INTO MSG-STRING  
END-STRING  
CALL 'CEEMOUT' USING MSG-STR, MSG-DEST, FEEDBACK  
CALL 'CEEMSG' USING CURRENT-CONDITION, MSG-DEST, FEEDBACK

EVALUATE TRUE

\*\*\*\*\*  
\* AN SOC1 HAS OCCURRED  
\* SET RESULT-CODE TO PERCOLATE TO ALLOW OTHER CONDITION  
\* HANDLERS TO GET CONTROL, IF ANY EXIST. OTHERWISE,  
\* DEFAULT LE CONDITION PROCESSING OCCURS, WHICH MAY  
\* TERMINATE THIS ENCLAVE.  
\*\*\*\*\*

WHEN CEE341  
DISPLAY 'ECH911 - ENTERED FOR OPERATION EXCP'  
MOVE 'Y' TO ERROR-INDICATOR  
SET PERCOLATE TO TRUE  
DISPLAY 'ECH911 - PERCOLATING AFTER OPERATION EXCP'

\*\*\*\*\*  
\* AN SOC4 HAS OCCURRED  
\* USE CEEMRCE TO SET THE RESUME CURSOR TO A PREVIOUSLY-  
\* DEFINED RESUME POINT.  
\* SET RESULT-CODE TO RESUME SO THAT CONTROL IS PASSED TO  
\* THE RESUME POINT.  
\*\*\*\*\*

WHEN CEE344  
DISPLAY 'ECH911 - ENTERED FOR PROTECTION EXCP'  
MOVE 'Y' TO ERROR-INDICATOR  
CALL 'CEEMRCE' USING RECOVER-ADDR, FEEDBACK  
IF FB-SEV = ZEROS  
DISPLAY 'ECH911 - CEEMRCE SUCCESSFUL'  
ELSE  
DISPLAY 'ECH911 - CEEMRCE FAILED'  
DISPLAY 'FB-MSG = ' FB-FAC, FB-MSG  
END-IF  
SET RESUME TO TRUE  
DISPLAY 'ECH911 - RESUMING AFTER PROTECTION EXCP'

\*\*\*\*\*  
\* AN SOCB HAS OCCURRED  
\* SET RESULT-CODE TO RESUME SO THAT CONTROL IS PASSED TO  
\* THE INSTRUCTION AFTER THE FAILING INSTRUCTION.  
\*\*\*\*\*

WHEN CEE34B  
DISPLAY 'ECH911 - ENTERED FOR DECIMAL DIVIDE EXCP'  
MOVE 'Y' TO ERROR-INDICATOR  
SET RESUME TO TRUE  
DISPLAY 'ECH911 - RESUMING AFTER DECIMAL DIVIDE EXCP'

\*\*\*\*\*  
\* A TABLE REFERENCE OUT OF BOUNDS ERROR HAS OCCURRED  
\* USE CEEMRCR TO SET THE RESUME CURSOR TO THE RETURN

```

*   POINT IN THE CALLER OF THE FAILING ROUTINE.
*   SET RESULT-CODE TO RESUME SO THAT CONTROL IS PASSED TO
*   THE RESUME POINT.
*****
      WHEN IGZ006
        DISPLAY 'ECH911  - ENTERED AFTER TABLE REF OOB'
        MOVE 'Y' TO ERROR-INDICATOR
        CALL 'CEEMRCR' USING MOVE-TYPE, FEEDBACK
        IF FB-SEV = ZEROS
          DISPLAY 'ECH911  - CEEMRCR SUCCESSFUL'
        ELSE
          DISPLAY 'ECH911  - CEEMRCR FAILED'
          DISPLAY 'FB-MSG = ' FB-FAC, FB-MSG
        END-IF
        SET RESUME TO TRUE
        DISPLAY 'ECH911  - RESUMING AFTER TABLE REF OOB'

*****
*   SOME OTHER CONDITION HAS OCCURRED
*   SET RESULT-CODE TO PERCOLATE TO ALLOW OTHER CONDITION
*   HANDLERS TO GET CONTROL, IF ANY EXIST. OTHERWISE,
*   DEFAULT LE CONDITION PROCESSING OCCURS, WHICH MAY
*   TERMINATE THIS ENCLAVE.
*****
      WHEN OTHER
        SET PERCOLATE TO TRUE
        DISPLAY 'ECH911  - CONDITION NOT RECOGNIZED, PERCOLATE'
      END-EVALUATE.

999-EXIT-PROGRAM.
      GOBACK.

END PROGRAM "ECH911".

```